1 REM ***  *****************************************************

1 REM ***  *              Baysix Automation Inc.               *

1 REM ***  *       Lift Station Level Controller Software      *

1 REM ***  *               Analog Version 1.0                  *

1 REM ***  *                by Kelly Torrance                  *

1 REM ***  *              Commissioned 06.12.92                *

1 REM ***  *                                                   *

1 REM ***  *         Copyright 1992 by                         *

1 REM ***  *         Baysix Automation Inc.                    *

1 REM ***  *         835 W. Grant Hwy.                         *

1 REM ***  *         Marengo Il. 60152                         *

1 REM ***  *         (815) 568-6037   FAX (815) 568-6932       *

1 REM ***  *                                                   *

1 REM ***  *      Unauthorized use or duplication of this      *

1 REM ***  *      software is strictly forbidden.              *

1 REM ***  *****************************************************

1 REM *** NUMERIC VARIABLES USED ***

1 REM A = HOUR-METER FOR MOTOR #1

1 REM B = HOUR-METER FOR MOTOR #2

1 REM C = HOUR-METER FOR MOTOR #3

1 REM D = SELECTS VFD OUTPUT TO MOTOR #1 OR #2 

1 REM E = TRUE EVERY OTHER SECOND

1 REM F = PREVENTS FLIP-FLOPS OF VARIABLE "D".

1 REM G = TRUE WHEN LEVEL GOES ABOVE SETPOINT "Q" FALSE WHEN BELOW S.P. "R"

1 REM H = TIMER FOR DISABLING MANUAL PUMP OPERATION

1 REM I = COUNTER FOR DELAYING HIGH LEVEL ALARM

1 REM J = COUNTER FOR DELAYING LOW LEVEL ALARM 

1 REM K = Tank level in inches

1 REM L = LCD DISPLAY MODE

1 REM M = PUMP 1 VFD IS ON

1 REM N = PUMP 1 LINE IS ON

1 REM O = PUMP 2 VFD IS ON

1 REM P = PUMP 2 LINE IS ON

1 REM Q = PUMP 3 IS ON

1 REM R = NOISE DAMPER FOR ANALOG INPUT (LEVEL)

1 REM S = SPEED OF THE VFD

1 REM T = Counter for displaying User-Entry-Error messages

1 REM U = 2-3 SECOND TIMER FOR DISPLAYING ERRORS

1 REM V = 5 MINUTE ALARM SILENCE TIMER

1 REM W = Reset for Variable "R" (1281,1283)

1 REM X = Temporary storage variable

1 REM Y = INITIALIZATION BIT

1 REM Z = Temporary storage variable

1 REM *** STRING VARIABLES USED ***

1 REM A$ = Operation Error Message

1 REM B$ = Operation Error Time

1 REM C$ = Command from user

1 REM D$ = Display Message

1 REM E$ = User-Entry-Error Message

1 REM F$ = Last Operation Error Message

1 REM G$ = 30 second timer to delay reporting Pump #1 VFD-not-on error

1 REM H$ = 30 second timer to delay reporting Pump #1 VFD-not-off error

1 REM I$ = 30 second timer to delay reporting Pump #1 line-not-on error

1 REM J$ = 30 second timer to delay reporting Pump #1 line-not-off error

1 REM K$ = 30 second timer to delay reporting Pump #2 VFD-not-on error

1 REM L$ = 30 second timer to delay reporting Pump #2 VFD-not-off error

1 REM M$ = 30 second timer to delay reporting Pump #2 line-not-on error

1 REM N$ = 30 second timer to delay reporting Pump #2 line-not-off error

1 REM O$ = 30 second timer to delay reporting Pump #3-not-on error

1 REM P$ = 30 second timer to delay reporting Pump #3-not-off error

1 REM Q$ =

1 REM R$ = 

1 REM S$ = Time and Date when program $tarted

1 REM T$ = Time and Date

1 REM U$ =

1 REM V$ =

1 REM W$ =

1 REM X$ = Active input$

1 REM Y$ = Active output$

1 REM Z$ = TEMPORARY STORAGE STRING

1 REM *** NON-VOLITAL RAM LOCATIONS USED ***

1 REM NVR 0 & 1 = A = HOUR-METER FOR MOTOR #1

1 REM NVR 2 & 3 = B = HOUR-METER FOR MOTOR #2

1 REM NVR 4 & 5 = C = HOUR-METER FOR MOTOR #3

1 REM NVR 6 = D = LAST MOTOR USED AS THE LEAD

1 REM NVR 7 = LEVEL IN INCHES THAT PUMP #3 WILL TURN ON

1 REM NVR 8 = LEVEL IN INCHES THAT PUMP #3 WILL TURN OFF

1 REM NVR 9 = LEVEL IN INCHES THAT LAG PUMP WILL TURN ON

1 REM NVR 10 = LEVEL IN INCHES THAT LAG PUMP WILL TURN OFF

1 REM NVR 11 = LEVEL IN INCHES THAT LEAD PUMP WILL TURN ON

1 REM NVR 12 = LEVEL IN INCHES THAT LEAD PUMP WILL TURN OFF

1 REM NVR 13 = WET WELL HIGH LEVEL

1 REM NVR 14 = WET WELL LOW LEVEL

1 REM NVR 15 = MINIMUM MOTOR SPEED IN PERCENT

1 REM NVR 16 = TANK LEVEL AT WHICH TO USE LOW SPEED

1 REM NVR 17 = # OF SECONDS OF LOW LEVEL BEFORE DISABLING MANUAL PUMPS

1 REM NVR 18 = 

1 REM NVR 19 = 

1 REM NVR 20 = 

1 REM NVR 21 = 

1 REM NVR 22 = 

1 REM NVR 23 = 

1 REM NVR 24 = 

1 REM NVR 25 =

7 REM *** DISABLE CONSOLE-BREAK ***

7 BYTE $001E = 0 : BYTE $007B = 0 : BYTE $007C = 0 : BYTE $002E = 0

1 REM *** SET AUX PORT TO 19200,N,8,1 ***

15 BYTE $C023 = 31 : BYTE $00AA = 5 : BYTE $00AB = 1

1 REM *** SET MAIN (PRINTER) SERIAL PORT TO 2400,N,8,1 ***

20 BYTE $C003 = 26 : BYTE $00A7 = 5 : BYTE $00A8 = 1

30 TRAP 7000 : CLRLCD : POSLCD 1

40 FOR Z = 1 TO 2

50 PRINT# Z "City of Woodstock Level Controller" ; : IF Z = 1 PRINT# Z " "

55 IF Z = 2 POSLCD 41

65 PRINT# Z "by Baysix Automation, Winter 1992" ; 

70 NEXT 

80 PRINT# 0 " " : PRINT# 1 " "

90 DBYTE 0 = 20

100 IF DBYTE 0 > 0 GOTO 100

110 F$ = "No errors recorded since last power up."

1 REM *** READ VARIABLES STORED IN NON-VOLITAL RAM TO HOUR-METER VARIABLES ***

200 A = NVR 0 * 256 + NVR 1

210 B = NVR 2 * 256 + NVR 3

220 C = NVR 4 * 256 + NVR 5

1 REM *** RECALL WHICH PUMP WAS LAST USED AS THE LEAD ***

230 D = NVR 6 : D = 1 - D

1 REM *** RECALL ON-OFF SETPOINTS FOR PUMPS ***

1 REM *** LAG LAG ON ***

250 IF (NVR 7 = 0) OR (NVR 7 > 100) NVR 7 = 80 

1 REM *** LAG LAG OFF ***

260 IF (NVR 8 = 0) OR (NVR 8 > NVR 7) NVR 8 = 30 

1 REM *** LAG ON ***

270 IF (NVR 9 = 0) OR (NVR 9 > NVR 7) NVR 9 = NVR 7 - 10

1 REM *** LAG OFF ***

280 IF (NVR 10 = 0) OR (NVR 10 > NVR 9) NVR 10 = 27

1 REM *** LEAD ON ***

290 IF (NVR 11 = 0) OR (NVR 11 > NVR 9) NVR 11 = NVR 9 - 10

1 REM *** LEAD OFF ***

300 IF (NVR 12 = 0) OR (NVR 12 > NVR 11) NVR 12 = 25 

1 REM *** RECALL WET WELL HI & LOW LEVELS ***

1 REM *** HIGH LEVEL ***

310 IF (NVR 13 < NVR 7) OR (NVR 13 > 110) NVR 13 = 110

1 REM *** LOW LEVEL ***

320 IF (NVR 14 > NVR 12) OR (NVR 14 > NVR 13) NVR 14 = NVR 12 - 1

1 REM *** LOW MOTOR SPEED % ***

330 IF (NVR 15>80) OR (NVR 15<40) NVR 15=50

1 REM *** LOW SPEED LEVEL ***

340 IF (NVR 16<=NVR 12) OR (NVR 16>=NVR 11) NVR 16 = NVR 12+1

1 REM *** SECONDS BEFORE DISABLING MANUAL ***

350 IF NVR 17 > 180 NVR 17 = 180

1 REM *** INITIALIZE RTC IF IT'S OUT OF RANGE ***

360 IF (BYTE 14 > 7) OR (BYTE 14 = 0) NVR-8 = 1 

365 IF (BYTE 15 > 31) OR (BYTE 15 = 0) NVR-7 = 1 

370 IF (BYTE 16 > 12) OR (BYTE 16 = 0) NVR-6 = 1 

375 IF BYTE 17 > 99 NVR-5 = 92 

380 IF BYTE 12 > 23 NVR-10 = 1 

385 IF BYTE 10 > 59 NVR-12 = 1 

390 IF BYTE 8 > 59  NVR-14 = 1 

1 REM *** SET 6-MINUTE TIMER TO INCREMENT MOTOR-USE COUNTERS ***

400 DBYTE 0 = 3600 : ONIRQ 0 , 9000 

410 REM *** SET 1 SECOND INTERUPT TIMER ***

410 DBYTE 2 = 10 : ONIRQ 1 , 10000

420 REM *** RESET LCD ***

420 CLRLCD

1 REM

1 REM

1 REM

1 REM

1 REM This routine updates the time-stamp-string for marking events.

1000 GOTO 1010 + BYTE 14

1010 GOTO 2000

1011 T$ = "Sun.," : GOTO 1020

1012 T$ = "Mon.," : GOTO 1020

1013 T$ = "Tue.," : GOTO 1020

1014 T$ = "Wed.," : GOTO 1020

1015 T$ = "Thu.," : GOTO 1020

1016 T$ = "Fri.," : GOTO 1020

1017 T$ = "Sat.,"

1020 GOTO 1030 + BYTE 16

1030 GOTO 2000

1031 T$ = T$ + "Jan." : GOTO 1050

1032 T$ = T$ + "Feb." : GOTO 1050

1033 T$ = T$ + "Mar." : GOTO 1050

1034 T$ = T$ + "Apr." : GOTO 1050

1035 T$ = T$ + "May " : GOTO 1050

1036 T$ = T$ + "June " : GOTO 1050

1037 T$ = T$ + "July " : GOTO 1050

1038 T$ = T$ + "Aug." : GOTO 1050

1039 T$ = T$ + "Sep." : GOTO 1050

1040 T$ = T$ + "Oct." : GOTO 1050

1041 T$ = T$ + "Nov." : GOTO 1050

1042 T$ = T$ + "Dec." : GOTO 1050

1050 T$ = T$ + STR$( BYTE 15 ) + ", "

1060 Z = BYTE 12

1070 IF Z = 0 T$ = T$ + "12:" : GOTO 1100

1080 IF Z < 13 T$ = T$ + STR$( Z ) + ":" : GOTO 1100

1090 T$ = T$ + STR$( Z - 12) + ":"

1100 IF BYTE 10 < 10 T$ = T$ + "0"

1110 T$ = T$ + STR$( BYTE 10 ) + ":" 

1120 IF BYTE 8 < 10 T$ = T$ + "0"

1130 T$ = T$ + STR$( BYTE 8 )

1140 IF Z < 13 T$ = T$ + " a.m.  " 

1150 IF Z > 12 T$ = T$ + " p.m.  "

1160 IF Y = 0 S$ = T$

1 REM *** BUILD INPUT$ ***

1200 IF L <> 1 GOTO 1260

1210 X$ = ""

1220 FOR Z = 1 TO 16

1230 IF PIN Z = 1 X$ = X$ + "1" : GOTO 1250

1240 X$ = X$ + "0"

1250 NEXT  

1 REM *** The following formula determines the level in the tank

1 REM *** based on a 4-20Ma input, a 250ohm input resistor, and 

1 REM *** a linear response from 0-160 inches.

1260 IF ADC 0 <= 51 K = 0 : GOTO 1281

1270 IF ADC 0 = 52 K = 1 : GOTO 1281

1 REM *** Analog to digital converters are 8 bit channels (0 to 255) ***

1280 K = ((ADC 0 - 51) * 100) / 127 : IF R = 0 R = K

1 REM *** This routine causes the analog input to be ignored if it jumps ***

1 REM *** up or down by 9 points in a single program scan.

1 REM *** Varable W is reset after 2 seconds in the 1 second interrupt ***

1 REM *** Varable R is reset along with W ***

1281 IF W > 0 K = R : GOTO 1290

1282 IF Y = 0 R = K : GOTO 1290

1283 IF (K<(R-9)) OR (K>(R+9)) K = R : W = 1 : GOTO 1290

1285 R = K

1 REM *** I is the HIGH LEVEL, J is the LOW...SEE 1840,1850 ***

1290 IF (K>NVR 13) AND (I<10) I = I + 1 : GOTO 1295

1292 I = 0

1295 IF (K<NVR 14) AND (J<10) J = J + 1 : GOTO 1300

1297 J = 0 

1300 GOTO 1400 + L

1 REM

1 REM

1 REM

1 REM

1400 D$ = T$ : GOTO 1500 

1401 D$ = "Inputs = " + X$ : GOTO 1500 

1402 D$ = "Outputs = " + Y$ + "   " : GOTO 1500

1403 D$ = "Pump # 1 has " + STR$( A/10 ) + " hours of use.       " : GOTO 1500 

1404 D$ = "Pump # 2 has " + STR$( B/10 ) + " hours of use.       " : GOTO 1500 

1405 D$ = "Pump # 3 has " + STR$( C/10 ) + " hours of use.       " : GOTO 1500 

1 REM *** CONVERT ANALOG INPUT VALUE TO MILLIVOLTS AND MILLIAMPS ***

1406 Z=(((ADC0*100)/51)*10):D$="Analog Input = "+STR$(Z)+"mV = "+STR$(Z/250)+"mA   ": GOTO 1500 

1 REM *** THE FORMULA FOR MOTOR SPEED IS BASED ON 2300 / 23 = % ***

1407 D$ = "Motor speed is presently " + STR$( S / 23 ) + "%  " : GOTO 1500 

1408 D$ = "Wet well level = " + STR$(K) + "''s. Last error = ": GOTO 1500 

1409 D$ = "LAG LAG turns ON at " + STR$( NVR 7 ) + " inches. ": GOTO 1500

1410 D$ = "LAG LAG turns OFF at " + STR$( NVR 8 ) + " inches. ": GOTO 1500

1411 D$ = "LAG Pump turns ON at " + STR$( NVR 9 ) + " inches. ": GOTO 1500

1412 D$ = "LAG Pump turns OFF at " + STR$( NVR 10 ) + " inches. ": GOTO 1500

1413 D$ = "LEAD Pump turns ON at " + STR$( NVR 11 ) + " inches. ": GOTO 1500

1414 D$ = "LEAD Pump turns OFF at " + STR$( NVR 12 ) + " inches. ": GOTO 1500

1415 D$ = "Wet well high level is " + STR$( NVR 13 ) + " inches. " : GOTO 1500  

1416 D$ = "Wet well low level is " + STR$( NVR 14 ) + " inches. " : GOTO 1500  

1417 D$ = "Low motor speed is " + STR$( NVR 15 ) + "% of full speed." : GOTO 1500  

1418 D$ = "Low speed is used below " + STR$( NVR 16 ) + " inches." : GOTO 1500  

1419 D$ = "Low Level Alarm timer = " + STR$( NVR 17 ) + " seconds." : GOTO 1500

1420 D$ = "(C) Baysix Electronics 1992          " : IF LEN( A$ ) <> 0 D$ = A$   

1 REM

1 REM

1 REM

1 REM

1500 POSLCD 1 : LCD D$ ; : POSLCD 41 

1510 IF LEN(E$)<>0 LCD E$ ; : T = T + 1 : GOTO 1600

1520 IF (L < 9) AND (L<>8) LCD "Command ? " ; C$ ; : IF LEN( C$ ) = 0 LCD "__________" ; 

1530 IF L = 8 LCD F$ ;

1540 IF (L > 8) AND (L<20) LCD "Press UP or DOWN arrow to change...     " ; 

1 REM *** DISPLAY ERROR MESSAGE TIME ***

1560 IF L = 20 LCD B$ ; : IF LEN(B$)=0 LCD "Booted " ; S$ ; 

1600 PRINT# 1 D$ ; "   Command ? " ; C$ ; "          " ; CHR$(13) ;

1610 IF T > 20 T = 0 : E$ = ""

1 REM

1 REM

1 REM

1 REM

1 REM *** ERROR MESSAGE & ALARM GENERATOR ***

1700 IF (PIN 11 = 1) AND (LEN(A$)<>0) F$ = A$ : A$ = "" : B$ = ""

1710 IF LEN( A$ ) <> 0 GOTO 2000

1720 IF LEN(G$)=10 A$ = "PUMP #1 VFD IS NOT ON !                 "

1730 IF LEN(H$)=10 A$ = "PUMP #1 VFD IS NOT OFF !                "

1740 IF LEN(I$)=10 A$ = "PUMP #1 LINE IS NOT ON !                "

1750 IF LEN(J$)=10 A$ = "PUMP #1 LINE IS NOT OFF !               "

1760 IF LEN(K$)=10 A$ = "PUMP #2 VFD IS NOT ON !                 "

1770 IF LEN(L$)=10 A$ = "PUMP #2 VFD IS NOT OFF !                "

1780 IF LEN(M$)=10 A$ = "PUMP #2 LINE IS NOT ON !                "

1790 IF LEN(N$)=10 A$ = "PUMP #2 LINE IS NOT OFF !               "

1 REM *** NEW LOGIC ADDED TO DELAY PUMP 3 FROM FALSE ERRORS 5-7-93 ***

1800 IF LEN(O$)=10 A$ = "PUMP #3 IS NOT ON !                     "

1810 IF LEN(P$)=10 A$ = "PUMP #3 IS NOT OFF !                    "

1820 IF PIN 12 = 1 A$ = "FREQUENCY DRIVE FAILURE !               "

1830 IF PIN 13 = 1 A$ = "COMPRESSOR FAILURE !                    "

1840 IF I => 9 A$ = "WET WELL LEVEL IS OVER " + STR$(NVR 13) + " INCHES !     "

1850 IF J => 9 A$ = "WET WELL LEVEL IS UNDER " + STR$(NVR 14) + " INCHES !     "

1860 IF (LEN( A$ ) <> 0) AND (LEN( B$ ) = 0) B$ = T$ : L = 20 : CLRLCD

1 REM *** SEND ERROR MESSAGES TO THE PRINTER  (MAIN PORT) ***

1870 IF (LEN( A$ ) <> 0) PRINT# 0 A$ ; "     " ; T$

1 REM *** CANCEL ALARM SILENCE IF A DIFFERENT ERROR OCCURS ***

1880 IF INSTR(A$,F$)=0 V = 0

1 REM

1 REM

1 REM

1 REM

1 REM *** MAIN CONTROL LOGIC ***

1 REM *** DEFINITION OF INPUTS ***

1 REM PIN 1 = FORCE PUMP #1 TO LEAD -\

1 REM PIN 2 = FORCE PUMP #2 TO LEAD ---> 3 POS. SWITCH

1 REM PIN 3 = PUMP 1 AUTO

1 REM PIN 4 = PUMP 2 AUTO

1 REM PIN 5 = PUMP 3 AUTO

1 REM PIN 6 = INPUT FROM AUX. CONTACT ON PUMP #1 CONTACTOR (LINE)

1 REM PIN 7 = INPUT FROM AUX. CONTACT ON PUMP #1 CONTACTOR (VFD)

1 REM PIN 8 = INPUT FROM AUX. CONTACT ON PUMP #2 CONTACTOR (LINE)

1 REM PIN 9 = INPUT FROM AUX. CONTACT ON PUMP #2 CONTACTOR (VFD)

1 REM PIN 10 = INPUT FROM AUX. CONTACT ON PUMP #3 CONTACTOR 

1 REM PIN 11 = ALARM ACKNOWLEDGE

1 REM PIN 12 = VFD FAULT

1 REM PIN 13 = COMPRESSOR FAULT

1 REM PIN 14 =

1 REM PIN 15 =

1 REM PIN 16 =

1 REM *** DEFINITION OF OUTPUTS ***

1 REM POUT 1 = CONNECT VFD TO PUMP 1

1 REM POUT 2 = CONNECT LINE TO PUMP 1

1 REM POUT 3 = SELECT VFD FOR PUMP 2

1 REM POUT 4 = SELECT LINE FOR PUMP 2

1 REM POUT 5 = SELECT LINE FOR PUMP 3

1 REM POUT 6 = ALARM LIGHT

1 REM POUT 7 = ALARM BUZZER

1 REM POUT 8 = ALARM RELAY

1 REM POUT 9 = DISABLE HAND OPERATION (NO-HAND) RELAY

1 REM POUT 10 = 

1 REM POUT 11 = 

1 REM POUT 12 = 

1 REM POUT 13 = 

1 REM POUT 14 = 

1 REM POUT 15 =

1 REM POUT 16 = On every-other second

1 REM *** "G" ALLOWS THE VFD TO TURN ON ***

1 REM NVR 11 = LEVEL IN INCHES THAT LEAD PUMP WILL TURN ON

1 REM NVR 12 = LEVEL IN INCHES THAT LEAD PUMP WILL TURN OFF

1 REM K = Tank level in inches

1 REM D = SELECTS VFD OUTPUT TO MOTOR #1 OR #2 

1 REM G = TRUE WHEN LEVEL GOES ABOVE SETPOINT "Q" FALSE WHEN BELOW S.P. "R"

1 REM M = PUMP 1 VFD IS ON

1 REM O = PUMP 2 VFD IS ON

2000 IF K => NVR 11 G = 1

2010 IF K <= NVR 12 G = 0

2020 IF ( M = 1 ) AND (PIN 2 = 1) G = 0

2030 IF ( O = 1 ) AND (PIN 1 = 1) G = 0

2080 M=0:O=0

2090 IF G = 0 GOTO 2180

2100 IF (PIN 3= 0) AND (PIN 4= 1) AND (D = 0) GOTO 2180

2110 IF (PIN 3= 1) AND (PIN 4= 0) AND (D = 1) GOTO 2180

1 REM *** Turn off pump 1 VFD ***

2120 IF PIN 3 = 0 M = 0 : GOTO 2150

1 REM

1 REM *** FORCE VFD PUMP #1 LOGIC ***

1 REM

2130 IF(PIN 1=1) F=1 : M=1 : N=0 : O=0 : GOTO 2180

1 REM

1 REM *** AUTO VFD PUMP #1 LOGIC ***

1 REM *** F = PREVENTS FLIP-FLOPS OF VARIABLE "D" *** 

1 REM

2140 IF (D=0) AND (PIN1=0) AND (PIN2=0) F=1 : M=1 : N=0 : O=0 : GOTO 2180

1 REM

1 REM *** BYPASS VFD PUMP #2 ***

1 REM

2150 IF PIN 4 = 0 O = 0 : GOTO 2180

1 REM

1 REM *** FORCE VFD PUMP #2 LOGIC ***

1 REM

2160 IF(PIN2=1) F=1 : O=1 : M=0 : P=0 : GOTO 2180

1 REM

1 REM *** AUTO PUMP 2 LOGIC ***

1 REM

2170 IF (D=1) AND (PIN1=0) AND (PIN2=0) F=1 : O=1 : M=0 : P=0

2180 IF (M=1) OR (O=1) GOTO 2190

2185 G=0 : M=0 : O=0 : IF F=1 NVR 6=D : F=0 : D=1-D 

1 REM

1 REM *** OUTPUT TO PUMP 1 LINE CONTACTOR ***

1 REM

2190 IF (PIN3=1) AND (M=0) AND (K>NVR 9) N = 1

2200 IF (PIN 3 = 0) OR (M=1) OR (K < NVR 10) N = 0

1 REM

1 REM *** OUTPUT TO PUMP 2 LINE CONTACTOR ***

1 REM

2210 IF (PIN 4 = 1) AND (O=0) AND ( K>NVR 9) P = 1 

2220 IF (PIN 4 = 0) OR (O=1) OR (K < NVR 10) P = 0

1 REM

1 REM *** OUTPUT TO PUMP 3 LINE CONTACTOR ***

2230 IF (PIN 5 = 1) AND (K > NVR 7) Q = 1 

2240 IF (PIN 5 = 0) OR (K < NVR 8) Q = 0

1 REM

1 REM *** This is the string displayed after OUTPUTS =

1 REM

2250 Y$ = STR$(M)+STR$(N)+STR$(O)+STR$(P)+STR$(Q)

2260 POUT 1 = M : POUT 2 = N : POUT 3 = O : POUT 4 = P : POUT 5 = Q

1 REM *** THIS ROUTINE DETERMINES THE VFD SPEED ***

1 REM *** IF NEITHER PUMP IS IN AUTO SET SPEED TO ZERO ***

2500 IF (PIN 3 = 0) AND (PIN 4 = 0) BELL 0 , 100 : S = 0 : GOTO 2750

1 REM *** IF PUMP 1 IS IN AUTO AND 2 IS NOT THEN IF THE #2 CONTACTOR IS ON

1 REM *** SET THE SPEED TO ZERO AND VICE-VERSA

2510 IF (PIN 3 = 1) AND (PIN 4 = 0) AND (PIN 9 = 1) BELL 0 , 100 : S = 0 : GOTO 2750

2520 IF (PIN 4 = 1) AND (PIN 3 = 0) AND (PIN 7 = 1) BELL 0 , 100 : S = 0 : GOTO 2750

1 REM *** IF BOTH PUMPS ARE IN AUTO BUT NIETHER CONTACTOR IS ON THEN

1 REM *** SET THE SPEED TO ZERO

1 REM *** THIS LOGIC WAS CHANGED FROM EXAMINING THE AUX. CONTACTS ON THE

1 REM *** VFD CONTACTORS TO THE M AND O VARIABLES THAT TURN THEM ON. 1-11-94

2530 IF (PIN 3 = 1) AND (PIN 4 = 1) AND (M = 0) AND (O = 0) BELL 0 , 100 : S = 0 : GOTO 2750

1 REM *** IF LEVEL IS > MINIMUM SEND SPEED REFERENCE TO THE VFD ***

1 REM *** THIS ASSUMES THAT 2300 PULSES PER SEC. = 20 ma. : 23 = 1% ***

1 REM *** USE MIN. SPEED IF LEVEL IS LESS THAN SETPOINT(NVR 16) ***

2600 IF K < NVR 16 S = NVR 15 * 23 : GOTO 2650

1 REM *** USE FULL SPEED IF LEVEL IS AT OR ABOVE SETPOINT(NVR 11) ***

2620 IF K => NVR 11 S = 2300 : GOTO 2650

1 REM *** Z = (100%-(USER#*1%))/(ON_SETPOINT-MIN_SPEED_SETPOINT) ***

2630 Z = (2300 - (NVR 15 * 23))/(NVR 11-NVR 16) 

1 REM *** Z = (LEVEL-MIN_SPEED_SETPOINT)*RESULT FROM 2630 ***

2640 Z = (K-NVR 16)*Z : S = ((NVR 15*23)+Z)

2650 BELL 20, S 

1 REM *** V DISABLES THE ALARMS FOR 360 SECONDS ***

1 REM *** AUDIBLE ALARM ( POUT 7 ) & CITY ALARM ( POUT 8 ) ***

2750 IF PIN 11 = 1 V = 360 : POUT 10 = 0

2760 IF V POUT 7 = 0 : POUT 8 = 0 : GOTO 2780

2770 IF LEN( A$ ) <> 0 POUT 7 = E : POUT 8 = 1 : POUT 10 = E

1 REM *** ALARM LIGHT ***

2780 IF (LEN( A$ ) <> 0) AND (E = 0) POUT 6 = 1 : GOTO 3000

2790 POUT 6 = 0

1 REM

1 REM

1 REM

1 REM

1 REM *** PARSE USER INPUT ***

3000 FOR Z = 1 TO 2

3005 IF Y = 0 DO : X = INCHR$( Z ) : UNTIL STAT# Z = 0 

3010 X = INCHR$( Z ) : IF X < 8 GOTO 3900 

1 REM *** Change lower case to upper case ***

3020 IF X > 90 X = X - 32

3030 IF X = 13 GOSUB 6000 : GOTO 3900

1 REM PRESSING [ESC] KEY CHANGES THE LCD DISPLAY AND CLEARS THE COMMAND$ 

1 REM PRESSING > KEY CHANGES LCD DISPLAY TO THE NEXT PAGE 

1 REM PRESSING < KEY CHANGES LCD DISPLAY TO THE PREVIOUS PAGE 

3040 IF (X = 8) AND (L > 0) L = L - 1 : C$ = "" : CLRLCD : GOTO 3060

3045 IF (X = 8) AND (L = 0) L = 20 : C$ = "" : CLRLCD

3050 IF X = 27 L = L + 1 : C$ = "" : CLRLCD

3055 IF X = 12  L = L + 1 : C$ = "" : CLRLCD

1 REM *** This line sets the number of displays possible ***

3060 IF L = 21 L = 0

1 REM *** PRESSING UP OR DOWN ARROWS CHANGES THE SETPOINTS ***

3070 IF L < 9 GOTO 3800

3080 IF X <> 11 GOTO 3300

3090 IF (L = 9) AND (NVR 7 < 159) NVR 7 = NVR 7 + 1 : GOTO 3700 

3095 IF L = 9 E$ = "Lag Lag : 159 inches maximum !          " : GOTO 3700 

3100 IF (L = 10) AND (NVR 8 < (NVR 7-1)) NVR 8 = NVR 8 + 1 : GOTO 3700

3105 IF L = 10 E$ = "Lag Lag : Off must be LOWER than On !   " : GOTO 3700 

3110 IF (L = 11) AND (NVR 9 < (NVR 7-1)) NVR 9 = NVR 9 + 1 : GOTO 3700

3115 IF L = 11 E$ = "Lag Pump : On must be LOWER than Lag Lag" : GOTO 3700 

3120 IF (L = 12) AND (NVR 10 < (NVR 9-1)) NVR 10 = NVR 10 + 1 : GOTO 3700

3125 IF L = 12 E$ = "Lag Pump : Off must be LOWER than On !  " : GOTO 3700 

3130 IF (L = 13) AND (NVR 11 < (NVR 9-1)) NVR 11 = NVR 11 + 1 : GOTO 3700

3135 IF L = 13 E$ = "Lead Pump : On must be LOWER than LAG ! " : GOTO 3700 

3140 IF (L = 14) AND (NVR 12 < (NVR 11-1)) NVR 12 = NVR 12 + 1 : GOTO 3700

3145 IF L = 14 E$ = "Lead Pump : Off must be LOWER than On ! " : GOTO 3700 

3150 IF (L = 15) AND (NVR 13 < 160) NVR 13 = NVR 13 + 1 : GOTO 3700 

3155 IF L = 15 E$ = "HIGH ALARM is 160 inches maximum !      " : GOTO 3700 

3160 IF (L = 16) AND (NVR 14 < (NVR 12-1)) NVR 14 = NVR 14 + 1 : GOTO 3700 

3165 IF L = 16 E$ = "LOW ALARM must be LESS than LEAD OFF !  " : GOTO 3700

3170 IF (L = 17) AND (NVR 15 < 80) NVR 15 = NVR 15 + 1 : GOTO 3700

3175 IF L = 17 E$ = "Low speed is 80% maximum !              " : GOTO 3700

3180 IF (L = 18) AND (NVR 16<(NVR 11-1)) NVR 16 = NVR 16 + 1 : GOTO 3700

3185 IF L = 18 E$ = "Low Speed Level must be < LEAD ON !" : GOTO 3700

3190 IF (L = 19) AND (NVR 17 < 180) NVR 17 = NVR 17 + 1 : GOTO 3700

3195 IF L = 19 E$ = "Low Level Alarm Delay is 180 MAXIMUM !  " : GOTO 3700

3300 IF X <> 10 GOTO 3700

3310 IF (L = 9) AND (NVR 7>(NVR 9+1)) NVR 7 = NVR 7 - 1 : GOTO 3700

3320 IF L = 9 E$ = "Lag Lag : On must be HIGHER than LAG !  " : GOTO 3700 

3330 IF (L = 10) AND (NVR 8>(NVR 10+1))NVR 8 = NVR 8 - 1 : GOTO 3700

3340 IF L = 10 E$ = "Lag Lag : Off must be HIGHER than LAG ! " : GOTO 3700 

3350 IF (L = 11) AND (NVR 9>(NVR 11+1)) NVR 9 = NVR 9 - 1 : GOTO 3700

3360 IF L = 11 E$ = "Lag Pump: On must be HIGHER than LEAD ! " : GOTO 3700 

3370 IF (L = 12) AND (NVR 10>(NVR 12+1)) NVR 10 = NVR 10 - 1 : GOTO 3700

3380 IF L = 12 E$ = "Lag Pump:Off must be HIGHER than LEAD ! " : GOTO 3700 

3390 IF (L = 13) AND (NVR 11>(NVR 12+1)) NVR 11 = NVR 11 - 1 : GOTO 3700

3400 IF L = 13 E$ = "Lead Pump: On must be HIGHER than Off ! " : GOTO 3700 

3410 IF (L = 14) AND (NVR 12>NVR 14 + 1) NVR 12 = NVR 12 - 1 : GOTO 3700

3420 IF L = 14 E$ = "Lead Pump : Off must be > LOW ALARM !   " : GOTO 3700 

3430 IF (L = 15) AND (NVR 13 > (NVR 7 + 1)) NVR 13 = NVR 13 - 1 : GOTO 3700 

3440 IF L = 15 E$ = "HIGH ALARM must be > than LAG LAG ON !  " : GOTO 3700 

3450 IF (L = 16) AND (NVR 14 > 0) NVR 14 = NVR 14 - 1 : GOTO 3700 

3460 IF L = 16 E$ = "LOW ALARM is zero inches minimum !      " : GOTO 3700

3470 IF (L = 17) AND (NVR 15>40) NVR 15 = NVR 15 - 1 : GOTO 3700

3480 IF L = 17 E$ = "Low speed is 40% minimum !              " : GOTO 3700

3490 IF (L = 18) AND (NVR 16>(NVR 12+1)) NVR 16 = NVR 16 - 1 : GOTO 3700

3500 IF L = 18 E$ = "Low Speed Level must be > LEAD OFF !    " : GOTO 3700

3510 IF (L = 19) AND (NVR 17 > 0) NVR 17 = NVR 17 - 1 : GOTO 3700

3520 IF L = 19 E$ = "Low Level Alarm Delay is 0 MINIMUM !    " : GOTO 3700

1 REM *** ERROR CHECK ***

3700 IF NVR 16 > (NVR 11 - 1) NVR 16 = NVR 11 - 1

3710 IF NVR 16 < (NVR 12 + 1) NVR 16 = NVR 12 + 1

3800 IF (X = 27) OR (X = 10) OR ( X = 13 ) X = 0 : GOTO 3900

1 REM *** IGNORE NON-ALPHA/NUMERIC CHARACTERS ***

3810 IF (X < 48) OR (X > 90) GOTO 3900

3820 IF (X > 57) AND (X < 65) GOTO 3900

3830 Z$ = CHR$( X ) : C$ = C$ + Z$

1 REM *** Limit the length of commands to 10 characters ***

3840 IF LEN( C$ ) => 11 C$ = ""

3850 IF STAT# Z <> 0 GOTO 3010

3900 NEXT

1 REM

5000 Y = 1 : GOTO 1000

1 REM

1 REM

1 REM

1 REM

1 REM *** PROCESS USER COMMAND ***

6000 IF INSTR(C$,"HOUR")=1 X=VAL(C$(5,2)): IF X < 24 NVR-10=X:C$ = ""

6020 IF INSTR(C$,"MINUTE")=1 X=VAL(C$(7,2)): IF X < 60 NVR-12=X:C$=""

6030 IF INSTR(C$,"DATE")=1 X=VAL(C$(5,2)): IF (X>0)AND(X<32) NVR-7=X:C$ = ""

6040 IF INSTR(C$,"DAY")=1 X=VAL(C$(4,1)): IF (X>0)AND(X<8) NVR-8=X:C$ = ""

6050 IF INSTR(C$,"MONTH")=1 X=VAL(C$(6,2)): IF (X>0)AND(X<13) NVR-6=X:C$ = ""

6060 IF INSTR(C$,"SYSTEM" ) = 1 GOSUB 12000 : STOP

6070 IF INSTR(C$,"RESET" ) = 1 GOSUB 12000 : ONIRQ 0,0 : C$ = "" : GOTO 1

6100 IF LEN( C$ ) <> 0 C$ = "Try again." 

6200 RETURN

1 REM

1 REM

1 REM

1 REM

1 REM *** ERROR TRAP ROUTINE ***

7000 PRINT# 1 "Error #" ; BYTE $001F ; " in Line #" ; DBYTE 32

7010 CLRLCD : LCD "Error #" ; BYTE $001F ; " in Line #" ; DBYTE 32 ;

7020 PRINT# 1 CHR$( 7 ) ;

1 REM *** TURN OFF ALL OUTPUTS BEFORE EXITING THE PROGRAM ***

7030 FOR X = 1 TO 16

7040 POUT X = 0

7050 NEXT

1 REM *** RECORD THE LEAD PUMP ***

7060 NVR 6 = D : G = 0 : A$ = "" : B$ = ""

7070 STOP

1 REM

1 REM

1 REM

1 REM

1 REM *** MOTOR-USE COUNTER INCREMENT SUBROUTINE ***

1 REM *** SET NVR 0-5 TO ZERO WHEN INSTALLED ***

1 REM *** THIS ROUTINE RUNS EVERY 6 MINUTES (1/10 hr.) ***

9000 IF (PIN 6 = 1) OR (PIN 7 = 1) A = A + 1

9010 IF (PIN 8 = 1) OR (PIN 9 = 1) B = B + 1

9020 IF (PIN 10 = 1) C = C + 1

9030 IF A > 65000 A = 0

9040 IF B > 65000 B = 0

9050 IF C > 65000 C = 0

9060 NVR 0 = A/256 : NVR 1 = A 

9070 NVR 2 = B/256 : NVR 3 = B 

9080 NVR 4 = C/256 : NVR 5 = C 

9100 DBYTE 0 = 3600 : RETURN

1 REM

1 REM

1 REM

1 REM

1 REM *** 1 SECOND INTERUPT ROUTINE ***

1 REM *** DECREMENT ALARM-SILENCE TIMER ***

10000 IF V V = V - 1

10010 IF K < NVR 14 H = H + 1 : GOTO 10030

10020 H = 0

10030 IF H > NVR 17 POUT 9 = 1 : GOTO 10050

10040 POUT 9 = 0

10050 E = 1 - E

10060 POUT 16 = E

1 REM *** RESET NOT-ACKNOWLEDGE MESSAGE

10070 IF INSTR( C$ , "Try" ) = 1 U = U + 1

10080 IF U = 3 U = 0 : C$ = ""

1 REM *** THE NEXT 16 LINES WERE ADDED TO DELAY PUMPS 1 AND 2 ON/OFF ERRORS 

1 REM *** BY 30 SECONDS. 12-31-93

1 REM *** PUMP 1 VFD NOT ON ***

10090 IF (M=1) AND (PIN 7= 0)  AND (LEN(G$)<10) G$ = G$ + "1" : GOTO 10110

10095 G$ = ""

1 REM *** PUMP 1 VFD NOT OFF ***

10100 IF (M=0) AND (PIN 7= 1) AND (PIN 3= 1)  AND (LEN(H$)<10) H$ = H$ + "1" : GOTO 10110

10105 H$ = ""

1 REM *** PUMP 1 LINE NOT ON ***

10110 IF (N=1) AND (PIN 6= 0)  AND (LEN(I$)<10) I$ = I$ + "1" : GOTO 10130

10115 I$ = ""

1 REM *** PUMP 1 LINE NOT OFF ***

10120 IF (N=0) AND (PIN 6= 1) AND (PIN 3= 1)  AND (LEN(J$)<10) J$ = J$ + "1" : GOTO 10130

10125 J$ = ""

1 REM *** PUMP 2 VFD NOT ON ***

10130 IF (O=1) AND (PIN 9= 0)  AND (LEN(K$)<10) K$ = K$ + "1" : GOTO 10150

10135 K$ = ""

1 REM *** PUMP 2 VFD NOT OFF ***

10140 IF (O=0) AND (PIN 9= 1) AND (PIN 4= 1)  AND (LEN(L$)<10) L$ = L$ + "1" : GOTO 10150

10145 L$ = ""

1 REM *** PUMP 2 LINE NOT ON ***

10150 IF (P=1) AND (PIN 8= 0)  AND (LEN(M$)<10) M$ = M$ + "1" : GOTO 10170

10155 M$ = ""

1 REM *** PUMP 2 LINE NOT OFF ***

10160 IF (P=0) AND (PIN 8= 1) AND (PIN 4= 1)  AND (LEN(N$)<10) N$ = N$ + "1" : GOTO 10170

10165 N$ = ""

1 REM *** THE NEXT 4 LINES WERE ADDED TO DELAY PUMP 3 ON/OFF ERRORS BY 30 SECONDS. 5-7-93

10170 IF (Q=1) AND (PIN 10 = 0) AND (LEN(O$)<10) O$ = O$ + "1" : GOTO 10210

10180 O$ = ""

10190 IF (Q=0) AND (PIN 10 = 1) AND (PIN 5 = 1) AND (LEN(P$)<10) P$ = P$ + "1" : GOTO 10210

10200 P$ = "" 

1 REM *** These lines were added to reset the analog input noise flag W ***

10210 IF W = 2 W = 0 : R = 0

10220 IF W = 1 W = 2

10230 DBYTE 2 = 10

10240 RETURN 

1 REM

1 REM

1 REM

1 REM

1 REM *** TURN OFF ALL OUTPUTS BEFORE EXITING THE PROGRAM ***

12000 FOR X = 1 TO 16

12010 POUT X = 0

12020 NEXT

1 REM *** RECORD THE LEAD PUMP ***

12030 NVR 6 = D : G = 0 : A$ = "" : B$ = ""

12040 PRINT# 1 CHR$(13) ; CHR$(10) ; "Program stopped by command." 

12050 CLRLCD : LCD "Program stopped by command." ; 

12100 RETURN

RUN

